# Maintainer: Bartłomiej Piotrowski <bpiotrowski@archlinux.org>

pkgname=rpcsvc-proto
pkgver=1.4
pkgrel=1
pkgdesc='rpcsvc protocol definitions from glibc'
arch=(x86_64)
url='https://github.com/thkukuk/rpcsvc-proto'
license=(BSD)
depends=(glibc)
source=($pkgname-$pkgver.tar.gz::https://github.com/thkukuk/rpcsvc-proto/archive/v$pkgver.tar.gz)
md5sums=('3b2724bb7d6560ddcb4dbcdb5465de10')

prepare() {
  cd $pkgname-$pkgver
  autoreconf -fi
}

build() {
  cd $pkgname-$pkgver
  ./configure --prefix=/usr
  make
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
  install -Dm644 COPYING "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}
